


# Linkage File for 2013 MEPS and 2011/2012 NHIS Public-use Files

## 1.0 Overview

The MEPS-HC survey uses the NHIS as its sampling frame. Each year a new MEPS-HC panel is established drawing from the previous year's NHIS sample. The MEPS-HC collects data through an overlapping panel design. In this design, two calendar years of information are collected from each household through in-person interviews, conducted over five rounds, with Round 3 spanning both calendar years.

For each panel, Rounds 1, 2, and part of Round 3 typically contain data from calendar year 1; the remaining part of Round 3, and Rounds 4 and 5 cover calendar year 2 for a panel. Therefore, MEPS full calendar year public-use data files contain data from the first year of a new panel combined with that of the second year of the previous panel. As illustrated below, for full calendar year 2013 estimates, Rounds 1, 2, and 3 of Panel 18 (which uses the 2012 NHIS as its sampling frame) are combined with Rounds 3, 4, and 5 of Panel 17 (which uses the 2011 NHIS as its sampling frame).



Public-use files containing NHIS data for a given calendar year are available from NCHS.

Users who need to augment the MEPS data with information from NHIS can do so with the linkage file described in the following sections.

## 2.0 Linkage File Description

The MEPS and NHIS linkage file, NHMEP13X.DAT, allows the data user to merge any of the person-level MEPS 2013 Full Year public use data files with the 2011 and 2012 NHIS person-level public-use data files (Person, Sample Adult, and Sample Child).

The NHIS person identifiers have been changed since 2004. Prior to 2004, one household equaled one "case" and unique person identifiers are Household Serial Number (HHX) and Person Number (PX). Since 2004, each family (FMX) is considered a separate case and unique person identifiers are Household Serial Number (HHX), Family Sequence Number (FMX), and Person Sequence Number (FPX).

The linkage file contains 36,940 person-level records and seven variables. In the linkage file, a record exists for each of the MEPS 2013 full-year persons. Each record contains the MEPS sample person ID (DUPERSID) and the corresponding NHIS sample person ID (HHX, FMX, and FPX). The linkage file can be linked to any of the person-level MEPS 2013 Full Year public use data files using the variable DUPERSID. The linkage file can be linked to the NHIS 2011 or 2012 person-level data files by HHX, FMX, FPX, and SRVY\_YR.

When a MEPS sample person does not link to NHIS, HHX is set to 999999, FMX is set to 99, PX is set to 99, FPX is set to 99, SRVY\_YR is set to 9999, and LINKFLAG is set to 0.

### 3.0 Linkage File Record Counts

16,660 of the 18,848 Panel 17 persons link to the 2011 NHIS data, while 16,439 of the 18,092 Panel 18 persons link to the 2012 NHIS data. A total of 3,841 persons in the two panels do not link to either 2011 or 2012 NHIS data. These unlinked cases include newborns, newly in-scope persons as well as a small number of cases where the NHIS identified a household as responding but when fielded in MEPS it was determined to actually be a non-responding household. The table below summarizes the linkages.

**Table 1 - Linkage File Record Counts**

	Linked to NHIS		Not Linked to NHIS	Total
	2011 NHIS PUF (n=101,875)	2012 NHIS PUF (n=108,131)		
<b>MEPS FY 2013 Panel 16 persons</b>	16,660	0	2,188	18,848
<b>MEPS FY 2013 Panel 17 persons</b>	0	16,439	1,653	18,092
<b>Total</b>	33,099		3,841	36,940

## 4.0 Linkage File Record Layout

Following is the record layout for the person-level MEPS/NHIS linkage file (NHMEP13X.DAT).

Variable	Column Position	Type	Label and value range
<b>DUPERSID</b>	1-8	Character	MEPS encrypted person ID (range=20004101-49683102)
<b>HHX</b>	9-14	Character	NHIS household serial number (range=000003 – 061138)
<b>FMX</b>	15-16	Character	NHIS family number (range=01- 08)
<b>FPX</b>	17-18	Character	NHIS person number (range=01-14)
<b>LINKFLAG</b>	19-19	Numeric	Linkage status between MEPS and NHIS (1 or 0)
<b>PANEL</b>	20-21	Numeric	MEPS panel number (either 17 or 18)
<b>SRVY_YR</b>	22-25	Numeric	NHIS survey year (2011 or 2012)

Below is the input statement to convert the linkage file (NHMEP13X.DAT) to a SAS dataset.

```
DATA XX.NHMEP12X;  
  INFILE "C:\TEMP\MEPS\NHMEP13X.DAT";  
  INPUT DUPERSID $1-8 HHX $9-14 FMX $15-16 FPX $17-18 LINKFLAG 19 PANEL 20-21 SRVY_YR 22-25;  
RUN;
```

## 5.0 Linking Instructions for SAS Users

The following is one way of adding NHIS person-level variables to the MEPS person-level file. Input files are: MEPS HC-157 (2013 Full-year Population Characteristics), 2011 NHIS person-level data file, 2012 NHIS person-level data file, and the linkage file NHMEP13X.DAT.

- (1) Create four SAS datasets as follows:
  - Convert MEPS HC-157 (ASCII or SAS transport file) to a SAS dataset named FY2013 (n=36,940)
  - Convert the linkage file NHMEP13X.DAT to a SAS dataset named NHMEP13X (n=36,940)
  - Convert the 2011 NHIS Person file to a SAS dataset named NHIS2011 (n=101,875). Make sure the SAS dataset includes HHX, FMX, FPX, SRVY\_YR, and other variables that are to be added to the MEPS full-year dataset.
  - Convert the 2012 NHIS Person file to a SAS dataset named NHIS2012 (n=108,131). Make sure the SAS dataset includes HHX, FMX, FPX, SRVY\_YR, and other variables that are to be added to the MEPS full-year dataset.
- (2) Sort FY2013 by DUPERSID. Concatenate NHIS2011 and NHIS2012 into one dataset named NHISALL (n=210,006). Sort NHISALL by HHX, FMX, FPX, and SRVY\_YR.
- (3) Merge FY2013 (n=36,940) with NHMEP13X (n=36,940) by DUPERSID. Name the output dataset

MEPS (n=36,940). Then sort MEPS by HHX, FMX, FPX, and SRVY\_YR.

- (4) Merge MEPS (n=36,940) with NHISALL (n=210,006) by HHX, FMX, FPX, and SRVY\_YR. Keep records only in MEPS (n=36,940). Name the output dataset MEPS13NH (n=36,940).

**Below is a sample SAS program for adding NHIS variables to the MEPS dataset.**

```
LIBNAME MEPS "C:\TEMP\MEPS";    /*MEPS 2013 Full-Year PUF */
LIBNAME NHIS "C:\TEMP\NHIS";    /*NHIS 2011 and 2012 Person Files*/

PROC FORMAT;
  VALUE AGE
    .='.'
    0-HIGH='>=0';
RUN;

DATA NHISALL;
  SET NHIS.NHIS2011 (KEEP=HHX FMX FPX SRVY_YR AGE_P /*other NHIS variables*/)
      NHIS.NHIS2012 (KEEP=HHX FMX FPX SRVY_YR AGE_P /*other NHIS variables*/);
RUN;

PROC SORT DATA=NHISALL;
  BY HHX FMX FPX SRVY_YR;
RUN;

DATA MEPS;
  MERGE MEPS.FY2013 MEPS.NHMEP13X (KEEP=DUPERSID HHX FMX FPX SRVY_YR LINKFLAG);
  BY DUPERSID;
RUN;

PROC SORT DATA=MEPS;
  BY HHX FMX FPX SRVY_YR;
RUN;

DATA MEPS.MEPS13NH;
  MERGE MEPS (IN=A) NHISALL;
  BY HHX FMX FPX SRVY_YR;
  IF A;
RUN;

TITLE1 "MEPS 2013 FY data with NHIS variables";
PROC FREQ DATA=MEPS.MEPS13NH;
  TABLES LINKFLAG*SRVY_YR*AGE_P/LIST MISSING;
  FORMAT AGE_P AGE.;
RUN;
```

## **6.0 Sample Stata Code for Adding NHIS Variables to the MEPS Dataset**

```
clear
```

```

log using stata13.log, replace

use c:\temp\mepps\fy2013
sort dupersid
save fy2013, replace

use c:\temp\nhis\nhis2011
append using c:\temp\nhis\nhis2012
save nhisall, replace

use nhisall
sort hhx fmx fpx srvy_yr
save nhisall, replace

use c:\temp\mepps\nhmep13x
sort dupersid
save link, replace

use fy2013
merge dupersid using link
drop _merge
save mepps, replace

use mepps
sort hhx fmx fpx srvy_yr
save mepps, replace

use mepps
merge hhx fmx fpx srvy_yr using nhisall
keep if _merge==3 | _merge==1
keep dupersid hhx fmx fpx srvy_yr linkflag /*plus nhis variables wanted to add*/
save c:\temp\mepps\mepps13nh, replace

describe
tab1 linkflag srvy_yr

```

## 7.0 Further Information

For any questions regarding the linkage file, please contact May Chu at 301-427-1653 or by email at [May.Chu@ahrq.hhs.gov](mailto:May.Chu@ahrq.hhs.gov). MEPS public use data files can be downloaded free of charge from the MEPS website at <http://www.mepps.ahrq.gov>. NHIS public use data files can be obtained by contacting NCHS by telephone (301-458-4636) or through their website, <http://www.cdc.gov/nchs>.